AD-A048 744 AIR FORCE FLIGHT DYNAMICS LAB WRIGHT-PATTERSON AFB OHIO F/G 9/2
NASTRAN STRESS RECOVERY PACKAGE.(U)
AUG 77 S E LAMBERSON

UNCLASSIFIED AFFDL-TR-77-84 NL

| OF |
ADA048 744

END
DATE FILMED
2-78
DDC

AFFDL-TR-77-84

NASTRAN STRESS RECOVERY PACKAGE

STEVEN E. LAMBERSON, 1 Lt, USAF

ANALYSIS AND OPTIMIZATION BRANCH
STRUCTURAL MECHANICS DIVISION

AUGUST 1977

TECHNICAL REPORT AFFDL-TR-77-84
Final Report for Period September 1976 to March 1977

Approved for public release; distribution unlimited.

AIR FORCE FLIGHT DYNAMICS LABORATORY
AIR FORCE WRIGHT AERONAUTICAL LABORATORIES
AIR FORCE SYSTEMS COMMAND
WRIGHT-PATTERSON AIR FORCE BASE, OHIO 45433

NOTICE

When Government drawings, specifications, or other data are used for any purpose other than in connection with a definitely related Government procurement operation, the United States Government thereby incurs no responsibility nor any obligation whatsoever; and the fact that the government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data, is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may in any way be related thereto.

This report has been reviewed by the Information Office (ASD/OIP) and is releasable to the National Technical Information Service (NTIS). At NTIS it will be available to the general public, including foreign nations.

This technical report has been reviewed and is approved for publication.

Steven E Lamberson

STEVEN E. LAMBERSON, 1 Lt, USAF
Project Engineer

Richard D Krobusek

RICHARD D. KROBUSEK, Maj, USAF
Chief, Analysis and Optimization Branch
Structural Mechanics Division

FOR THE COMMANDER:

Howard L Farmer

HOWARD L. FARMER, COL, USAF
Chief, Structural Mechanics Division
AF Flight Dynamics Laboratory

Copies of this report should not be returned unless return is required by security considerations, contractual obligations, or notice on a specific document.

AIR FORCE/56780/16 December 1977 — 200

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE	READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER: AFFDL-TR-77-84	2. GOVT ACCESSION NO. / 3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle): NASTRAN STRESS RECOVERY PACKAGE	5. TYPE OF REPORT & PERIOD COVERED: rept. FINAL, Sep 76 - Mar 77
	6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s): Steven E. Lamberson, 1 Lt, USAF	8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS: Air Force Flight Dynamics Laboratory, Air Force Wright Aeronautical Laboratories, Air Force Systems Command, Wright-Patterson Air Force Base, Ohio 45433	10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: 62201F; 1467, 146702; 14670236
11. CONTROLLING OFFICE NAME AND ADDRESS: Analysis and Optimization Branch (AFFDL/FBR), Structural Mechanics Division, Air Force Flight Dynamics Laboratory	12. REPORT DATE: Aug [redacted]77
	13. NUMBER OF PAGES: 21 (12) 39p.
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)	15. SECURITY CLASS. (of this report): Unclassified
	15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Finite Element Analysis, Postprocessors, Element Stresses.

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This user's manual presents the stress recovery package. This series of programs allows automated scanning of NASTRAN element stresses. This package can be used interactively or as part of a NASTRAN batch run. The package is operational on the CDC 6000 series computers.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

Ø12 Ø7Ø

FOREWORD

This report was prepared by Lt Steven Lamberson of the Structural Mechanics Division of the Air Force Flight Dynamics Laboratory. This work was performed under Project 1467 "Design and Analysis Methods for Aerospace Vehicle Structures", Task 146702, "Design and Analysis-Methods for Aerospace Structures", Work Unit 14670236, "NASTRAN Computer Program."

This report was released by the author in August 1977.





TABLE OF CONTENTS

SECTION I

INTRODUCTION

When performing large finite element analysis it is often desirable to use a postprocessor to identify those elements within the structure which are most critically loaded. The stress recovery package described in this report identifies those elements within the structure having safety margins less than a user specified value. This package can be used as a part of the initial NASTRAN analysis run or interactively at some later time.

This package is composed of three programs. Program STRESSA extracts the stress data from the NASTRAN output. Program YIELD uses the NASTRAN data deck to generate failure stress data for membrane and plate elements. Program STRESSB uses the data generated by STRESSA and YIELD to calculate safety margins for the elements and identify those elements with safety margins less than a user specified value. These programs are currently operational on the CDC 6000 series computers. The programs are coded in FORTRAN Extended; however, some modifications may be required for use with other computer systems.

SECTION II

The NASTRAN Run

Several changes must be made when running NASTRAN with this package. The TABPT module in NASTRAN allows data blocks to be stored in a compressed format on the output file. In order to use this feature a call to TABPT must be placed in the DMAP sequence. This call should be after the SAVE CARDNO $ command following the call to SDR2 (the stress recovery package). This can be accomplished by placing an Alter Package in the Executive Control Deck such as the following (for Rigid Format 1,0):

```
ALTER 121
TABPT OES1,,,, // $
ENDALTER
```

NASTRAN only calculates element stresses when stress output is requested in the Case Control Deck. This can be done either with a STRESS=ALL or STRESS(PUNCH)=ALL command. These commands allow the stresses to be calculated with or without printed stress output respectively.

The following NASTRAN load and execute command should be used: NASTRAN (DATA1,DATA2,DUMMY)ATTACH. The local file DATA1 contains the NASTRAN bulk data deck to be executed. DATA2 will contain the entire NASTRAN output file from which the compressed stress table printed by TABPT will be removed in STRESSA. DUMMY contains the stress data requested with the STRESS (PUNCH)=ALL card and can be discarded. The file DATA1 must be at the beginning of information when NASTRAN is executed. All the other data files used by the programs in this package except the input file to STRESSB are rewound as required internally.

SECTION III

STRESSA

The program STRESSA reads the NASTRAN output file, removes the stress table generated with the TABPT module, and places the remainder on the system output file. A data file (TAPE3) is generated containing the stress information in the format used by STRESSB. If the user wishes to use the stress recovery package interactively rather than as a part of the NASTRAN run the program STRESSA should be executed after the NASTRAN execution and TAPE3 stored for later use.

SECTION IV

YIELD

NASTRAN does not calculate safety margins for membrane and plate elements. Program STRESSB (SUBROUTINE FAILURE) generates these safety margins using failure stress data.

Program YIELD uses the NASTRAN data deck to generate failure stress data for the membrane and plate elements based on values given on MAT1 and MAT2 cards. These values are then used by program STRESSB along with the stresses from STRESSA to determine the safety margin which is not calculated by NASTRAN for these elements. YIELD currently handles the following element types

CQDMEM	CTRIA1	CQUAD1
CTRMEM	CTRIA2	CQUAD2
CTRBSC	CTRPLT	CQDPLT

Some requirements are placed on the NASTRAN data file for use by program YIELD. Card identifiers (col. 1-8) must be left justified for element cards, property cards, and MAT1 and MAT2 cards. Exponents used in yield stress values on MAT1 and MAT2 cards must be right justified. Continuation cards for MAT1 and MAT2 cards must immediately follow their parents.

SECTION V

STRESSB

Program STRESSB lists all elements whose safety margins are below a user specified value. Output for these elements consists of the element number, element type and the stresses in that element. All stresses normally output by NASTRAN are output for these elements. STRESSB will process the following element types

CROD	CONROD	CTRIA1	CQDMEM
CBEAM	CELAS1	CTRBSC	CTRIA2
CTUBE	CELAS2	CTRPLT	CQUAD2
CSHEAR	CELAS3	CTRMEM	CQUAD1
CTWIST	CELAS4	CQDPLT	
CBAR			

Where possible STRESSB uses safety margins calculated by NASTRAN. The safety margins of plate and membrane elements are determined in subroutine FAILURE. This subroutine checks the major and minor principal stresses against the tensile or compressive failure stresses according to their sign. FAILURE also compares the maximum shear stress with the shear failure stress. If any of these are below the user specified value the element is listed. For bending elements this is done for both surfaces. The user may supply his own subroutine FAILURE. Elements not included in the above list and elements for which no failure stresses have been supplied will be ignored by STRESSB.

STRESSB examines the element stresses by subcase. When the program is executed interactively, safety margin and subcase number combinations

are input directly from the terminal when requested by the program. When STRESSB is executed as part of the NASTRAN analysis run it is necessary to use an input file with two card images for each subcase to be examined; one card image contains the real value of the safety margin and the integer value of the subcase number separated by a comma, the other card image contains the word yes or no indicating whether there is or is not a subsequent subcase to be examined.

Some users may wish to input the failure stress tables directly rather than use program YIELD. This can be done by providing TAPE4 and TAPE5. TAPE4 must contain one card image for each membrane or plate element to be examined containing the Element Number and the Failure Set Identification Number of the appropriate Failure Stress Set (2I5). TAPE5 must contain one card image for each set of unique Failure Stresses containing the Failure Set Identification Number and the Failure Stresses in Tension, Compression, and Shear (I5,3F13.0).

APPENDIX A

PROGRAM STRESSA

Purpose: To recover the compressed stress data from a NASTRAN output prior to printing.

External References: None

```
*DECK STRESSA
      PROGRAM STRESSA(INPUT,OUTPUT,TAPE1=INPUT,TAPE2=OUTPUT,TAPE3
     1,TAPE6,TAPE7)
      DIMENSION A(14),C(10),D(10),IV(50)
      REWIND 1
      III=0

C        THIS SUBROUTINE STRIPS THE
C        NASTRAN OUTPUT FILE(INPUT),
C        PLACES THE STRESS DATA ON A
C        DATA FILE(TAPE3),AND THE
C        REMAINDER OF THE NASTRAN OUTPUT
C        ON THE SYSTEM OUTPUT FILE(OUTPUT),
C
C        DIMENSIONS IN THIS PROGRAM
C        ARE NOT PROBLEM DEPENDENT

1     READ(1,2)A
2     FORMAT(A1,13A10)
      IF(EOF(1))1000,3
3     IF(A(2).EQ.10HRECORD NO.)GO TO 10
      WRITE(2,2)A
      GO TO 1
10    III=1
      IOUT=3
11    READ(1,2)A
      IF(A(2).NE.10HRECORD NO.)GO TO 11
      IF(III.EQ.2)GO TO 150
101   READ(1,102)ISK
102   FORMAT(A1)
      IF(EOF(1))100,103
103   IF(ISK.NE.1H )GO TO 101
      BACKSPACE 1
      READ(1,12)ITYPE,ISUB,NWDS
12    FORMAT(27X,2I13,65X,I13)
      IF(ISUB.EQ.0)GO TO 101
      WRITE(IOUT,18)ITYPE,ISUB,NWDS
18    FORMAT(10HNEW RECORD,3I10)
13    READ(1,14)B
14    FORMAT(1X,A10)
      IF(B.NE.10HRECORD NO.)GO TO 13
150   IF(III.EQ.2)WRITE(IOUT,18)
15    READ(1,16)DDD,(C(I),D(I),I=1,10)
16    FORMAT(A1,10(A10,A3))
      IF(DDD.EQ.1H1)GO TO 70
      IF(C(1).EQ.10HRECORD NO.)GO TO 50
      DO 20 I=1,10
      IF(C(I).EQ.10H           .AND.D(I).EQ.3H   )GO TO 15
      WRITE(IOUT,17)C(I),D(I)
17    FORMAT(10X,A10,A3)
20    CONTINUE
      GO TO 15
50    BACKSPACE 1
      GO TO 11
70    READ(1,71)
71    FORMAT(10X)
      READ(1,71)
```

```
      READ(1,71)
      GO TO 15
100   WRITE(IOUT,18)
      GO TO 1
1000  STOP
      END
```

APPENDIX B

PROGRAM YIELD

Purpose: This program generates failure stress data for NASTRAN membrane and plate values from data available on MAT1 and MAT2 cards.

Arrays: IPROP (NPROP) where NPROP is the largest membrane or plate property number.

External References: RIJUST

```
*DECK YIELD
      PROGRAM YIELD(INPUT,OUTPUT,TAPE1=INPUT,TAPE2=OUTPUT,TAPE4,TAPE5,
     1TAPE6)
      DIMENSION A(12),ELEM(9),IPROP(2000)
      INTEGER RIJUST
C
C       IPROP SHOULD BE DIMENSIONED AS FOLLOWS:
C
C
C       IPROP(NPROP) WHERE NPROP = THE LARGEST MEMBRANE
C       OR PLATE PROPERTY NUMBER

      ELEM(1)=7HQDMEM
      ELEM(2)=7HTRMEM
      ELEM(3)=7HTRIA1
      ELEM(4)=7HTRBSC
      ELEM(5)=7HTRPLT
      ELEM(6)=7HQDPLT
      ELEM(7)=7HTRIA2
      ELEM(8)=7HQUAD2
      ELEM(9)=7HQUAD1
      DO 10  JJ=1,9
      REWIND 1
    1 READ(1,2)A
    2 FORMAT(A1,A7,8A8,A1,A7)
      IF(A(2).EQ.7HNDDATA )GO TO 5
      IF(A(2).NE.ELEM(JJ))GO TO 1
      IF(A(1).EQ.1HP)GO TO 4
      IELEM=RIJUST(A(3))
      IPRO1=RIJUST(A(4))
      WRITE(6,3)IELEM,IPRO1
    3 FORMAT (2I5)
      GO TO 1
    4 IPROP(RIJUST(A(3)))=RIJUST(A(4))
      GO TO 1
    5 REWIND 6
    6 READ(6,3)IELEM,IPRO1
      IF(IELEM.EQ.0)GO TO 8
      IF(EOF(6))8,7
    7 WRITE(4,3)IELEM,IPROP(IPRO1)
      GO TO 6
    8 REWIND 6
      WRITE(6,21)
21    FORMAT(10X)
      REWIND 6
      DO 9 J=1,2000
    9 IPROP(J)=0
   10 CONTINUE
      REWIND 1
   11 READ(1,2)A
      IF(A(2).EQ.7HNDDATA )GO TO 20
      IF(A(2).EQ.7HAT1    )GO TO 12
      IF(A(2).EQ.7HAT2    )GO TO 18
      GO TO 11
   12 J=RIJUST(A(3))
      READ(1,13)TEST
   13 FORMAT(1X,A7)
```

```
      IF(TEST.NE.A(12))GO TO 16
      BACKSPACE 1
      READ(1,14)X,Y,Z
   14 FORMAT(8X,3F8.0)
   17 WRITE(5,15)J,X,Y,Z
   15 FORMAT(I5,3F13.2)
      GO TO 11
   16 X=0.0
      Y=0.0
      Z=0.0
      GO TO 17
   18 J=RIJUST(A(3))
      READ(1,13)TEST
      IF(TEST.NE.A(12))GO TO 16
      BACKSPACE 1
      READ(1,19)X,Y,Z
   19 FORMAT(48X,3F8.0)
      GO TO 17
   20 STOP
      END
```

FUNCTION RIJUST

Entry Point: RIJUST

Purpose: To convert an alpha variable to integer format with right justification.

Calling Sequence: I1=RIJUST(IWORD)

IWORD = An alpha format of an integer variable

I1 - The right justified integer

```
      INTEGER FUNCTION RIJUST(IWORD)
      MASK=55B
      DO 10 I=1,10
      ITEST=77B.AND.IWORD
      IF(MASK.NE.ITEST)GO TO 20
10    IWORD=SHIFT(IWORD,54)
20    DECODE(10,30,IWORD)RIJUST
30    FORMAT(I10)
      RETURN
      END
```

APPENDIX C

PROGRAM STRESSB

Purpose: This program examines the stress data from a NASTRAN execution and prints out those elements for which the safety margin is below a user specified value.

Blank Common:

ISAFT(NSAFT),SAFMAR,SAFTVAL(NVAL3),X(20)

NSAFT = The largest membrane or plate element number.

ISAFT(NN) = The property number of plate or membrane element NN.

NVAL = The largest property number of a plate or membrane element.

SAFTVAL(NN,J) = The J'th failure stress for elements with property number NN (tension, compression, and shear).

SAFMAR = The user input critical safety margin.

X(J) - The stresses in the current element.

External References: CROD,CSHEAR,CMEM,ELAS,CBEAM

```
*DECK STRESSB
      PROGRAM STRESSB(INPUT,OUTPUT,TAPE1=INPUT,TAPE2=OUTPUT,
     1TAPE3,TAPE4,TAPE5)
      COMMON ISAFT(10000),SAFMAR,SAFTVAL(1000,3),X(20)
C
C        THE VARIABLES SHOULD BE DIMENSIONED AS FOLLOWS
C
C         ISAFT(NSAFT)
C        SAFTVAL(NVAL,3)
C
C        NSAFT=THE LARGEST MEMBRANE OR PLATE ELEMENT NUMBER
C        NVAL=THE LARGEST MEMBRANE OR PLATE PROPERTY NUMBER
C
      REWIND 1
      REWIND 4
1     READ(4,2)I,ISAFT(I)
2     FORMAT(2I5)
      IF(EOF(4))33,1
33    REWIND 5
34    READ(5,35)I,(SAFTVAL(I,J),J=1,3)
35    FORMAT(I5,3E13.6)
      IF(EOF(5))31,34
31    WRITE(2,4)
4     FORMAT(" WHAT IS THE CRITICAL SAFETY MARGIN AND SUBCASE NUMBER?")
      READ(1,*)SAFMAR,INUMB
      REWIND 3
5     READ(3,6)A
6     FORMAT(A10,3I10)
      IF(A.NE.10HNEW RECORD)GO TO 5
      BACKSPACE 3
      READ(3,6)A,ITYPE,ISUB,NWDS
      IF(ITYPE.EQ.0)GO TO 100
      IF(ISUB.NE.INUMB)GO TO 5
      IF(ITYPE.EQ.1)CALL CROD
      IF(ITYPE.EQ.2)CALL CBEAM(ITYPE)
      IF(ITYPE.EQ.3)CALL CROD
      IF(ITYPE.EQ.4)CALL CSHEAR
      IF(ITYPE.EQ.5)CALL CSHEAR
      IF(ITYPE.EQ.6)CALL CPLATE
      IF(ITYPE.EQ.7)CALL CPLATE
      IF(ITYPE.EQ.8)CALL CPLATE
      IF(ITYPE.EQ.9)CALL CMEM
      IF(ITYPE.EQ.10)CALL CROD
      IF(ITYPE.EQ.11)CALL ELAS
      IF(ITYPE.EQ.12)CALL ELAS
      IF(ITYPE.EQ.13)CALL ELAS
      IF(ITYPE.EQ.14)GO TO 5
      IF(ITYPE.EQ.15)CALL CPLATE
      IF(ITYPE.EQ.16)CALL CMEM
      IF(ITYPE.EQ.17)CALL CPLATE
      IF(ITYPE.EQ.18)CALL CPLATE
      IF(ITYPE.EQ.19)CALL CPLATE
      IF(ITYPE.GT.19.AND.ITYPE.LT.34)GO TO 5
      IF(ITYPE.EQ.34)CALL CBEAM(ITYPE)
      IF(ITYPE.GT.34)GO TO 5
      GO TO 5
100   WRITE(2,101)
```

```
101   FORMAT(" ARE ANY MORE SUBCASES TO BE EXAMINED?")
      READ(1,102)C
102   FORMAT(A1)
      IF(C.EQ.1HY)GO TO 1
      STOP
      END
```

SUBROUTINE CROD

Entry Point: CROD

Purpose: To determine if a rod's element stresses are below the critical value.

Calling Sequence:

```
CALL CROD
```

Blank Common: See main program

X(J) not used

```
      SUBROUTINE CROD
      COMMON ISAFT(10000),SAFMAR,SAFTVAL(1000,3)
      WRITE(2,3)
3     FORMAT(1H1," ROD ELEMENTS"/"     STRESS     SAFETY MARGIN")
6     READ(3,1)A,IELEM
1     FORMAT(A10,I12)
      IF(A.EQ.10HNEW RECORD)GO TO 10
      READ(3,2)AXIAL
2     FORMAT(10X,F13.0)
      READ(3,2)ASM
      IF(ASM.EQ.1.0)GO TO 20
      IF(ASM.GT.SAFMAR)GO TO 20
      WRITE(2,4)IELEM,AXIAL,ASM
4     FORMAT(1X,I10,E13.6,F13.8," AXIAL")
20    READ(3,2)AXIAL
      READ(3,2)ASM
      IF(ASM.EQ.1.0)GO TO 6
      IF(ASM.GT.SAFMAR)GO TO 6
      WRITE(2,7)IELEM,AXIAL,ASM
7     FORMAT(1X,I10,E13.6,F13.8," TORSIONAL")
      GO TO 6
10    BACKSPACE 3
      RETURN
      END
```

SUBROUTINE CSHEAR

Entry Point: CSHEAR

Purpose: To determine if the stress in a shear panel is below the critical level.

Calling Sequence

```
CALL CSHEAR
```

Blank Common: See main program

X(1) = Maximum Shear Stress

X(2) = Average Shear Stress

X(3) = Safety Margin

```
      SUBROUTINE CSHEAR
      COMMON ISAFT(10000),SAFMAR,SAFTVAL(1000,3),X(3)
      WRITE(2,3)
3     FORMAT(1H1," SHEAR PANELS"/
     1" ELEMENT    MAX SHEAR     AVE SHEAR   SAFETY  MARGIN")
1     READ(3,2)A,IELEM
2     FORMAT(A10,I12)
      IF(A.EQ.10HNEW RECORD)GO TO 10
      DO 6 I=1,3
      READ(3,4)X(I)
4     FORMAT(10X,F13.0)
6     CONTINUE
      IF(X(3).EQ.1.0)GO TO 1
      IF(X(3).GT.SAFMAR)GO TO 1
      WRITE(2,7)IELEM,X
7     FORMAT(I6,3E13.6)
      GO TO 1
10    BACKSPACE 3
      RETURN
      END
```

SUBROUTINE CMEM

Entry Point: CMEM

Purpose: To determine if the stress field in a membrane element is critical.

External References: FAILURE

Calling Sequence:

```
CALL CMEM
```

Blank Common: See Main Program

X(1) = not used

X(2) = Normal-X

X(3) = Normal-Y

X(4) = Shear-XY

X(5) = 0-Shear Angle

X(6) = Major Principal

X(7) = Minor Principal

X(8) = Maximum Shear

```
      SUBROUTINE CMEM
      COMMON ISAFT(10000),SAFMAR,SAFTVAL(1000,3),X(8)
      WRITE(2,3)
3     FORMAT(1H1," MEMBRANE ELEMENTS"/" ELEMENT NUMBER"/
     117X,"NORMAL-X",6X,"NORMAL-Y",6X,"SHEAR-XY",7X,
     2"ANGLE",7X,"MAJPRIN",6X,"MINPRIN",2X,"MAX-SHEAR")
1     READ(3,2)A,IELEM
2     FORMAT(A10,I12)
      IF(A.EQ.10HNEW RECORD)GO TO 10
      DO 6 I=2,8
      READ(3,4)X(I)
4     FORMAT(10X,F13.0)
6     CONTINUE
      IT=1
      CALL FAILURE(IELEM,IT)
      GO TO 1
10    BACKSPACE 3
      RETURN
      END
```

SUBROUTINE FAILURE

Entry Point: FAILURE

Purpose: To calculate safety margins for plate and membrane elements.

Calling Sequence:

CALL FAILURE (IELEM,IT)

IELEM = the element number

IT = 1 for membrane elements

= 2 for plate elements

Blank Common: See main program

X(J) see CMEM or CPLATE

```
      SUBROUTINE FAILURE(IELEM,IT)
      COMMON ISAFT(10000),SAFMAR,SAFTVAL(1000,3),X(16)
C
C
C        THIS SUBROUTINE DEFINES THE FAILURE CRITERIA
C        FOR PLATE AND MEMBRANE ELEMENTS
C
C
C        THE USER MAY REPLACE THE CODE BETWEEN HERE AND STATEMENT
C        20 WITH HIS OWN ALGORITHM TO EXAMINE THE STRESS
C        FIELD AND DETERMINE WHETHER TO PRINT OUT THE STRESSES
C        FOR THE ELEMENT(GO TO 20) OR NOT(GO TO 30).  THE FOL-
C        LOWING VARIABLES ARE AVAILABLE FOR USE
C
C        IELEM=ELEMENT NUMBER
C        IT=1 FOR MEMBRANE ELEMENTS
C        IT=2 FOR PLATE ELEMENTS
C        ISAFT(IELEM)=THE FAILURE SET ID FOR ELEMENT IELEM
C        SAFMAR =THE USER SPEC IFIED C RITIC AL SAFETY MARGIN
C        SAFTVAL(ISAFT(IELEM),1) =TENSILE FAILURE STRESS
C                                 2=COMPRESSIVE FAILURE STRESS
C                                 3= SHEAR FAILURE STRESS
C                    MEMBRANES                    PLATES
C          X(1)       0.0                    Z1 FIBRE DISTANCE 1
C            2        NORMAL-X               NORMAL-X AT Z1
C            3        NORMAL-Y               NORMAL-Y AT Z1
C            4        SHEAR-XY               SHEAR-XY AT Z1
C            5        0-SHEAR ANGLE          0-SHEAR ANGLE AT Z1
C            6        MAJ. PRIN.              MAJ.PRIN. AT Z1
C            7        MIN. PRIN.              MIN.PRIN. AT Z1
C            8        MAX. SHEAR              MAX.SHEAR AT Z1
C            9        NOT USED                Z2 FIBRE DISTANCE 2
C           10        NOT USED                NORMAL-X AT Z2
C           11        NOT USED                NORMAL-Y AT Z2
C           12        NOT USED                SHEAR-XY AT Z2
C           13        NOT USED                0-SHEAR ANGLE AT Z2
C           14        NOT USED                MAJ.PRIN. AT Z2
C           15        NOT USED                MIN.PRIN. AT Z2
C           16        NOT USED                MAX. SHEAR AT Z2
      IF(ISAFT(IELEM).EQ.0)RETURN
      SAFETYT=SAFTVAL(ISAFT(IELEM),1)
      SAFETYC=-SAFTVAL(ISAFT(IELEM),2)
      SAFETYS=SAFTVAL(ISAFT(IELEM),3)
      N=1
1     NN=N+5
      NNN=N+6
      DO 3 I=NN,NNN
      IF(X(I).EQ.0)GO TO 3
      IF(X(I).GT.0.0)GO TO 7
      SAFE=SAFETYC/X(I)
      IF(SAFE.LE.SAFMAR)GO TO 20
      GO TO 3
7     SAFE=SAFETYT/X(I)
      IF(SAFE.LE.SAFMAR)GO TO 20
3     CONTINUE
      IF(X(I).EQ.0)RETURN
      NN=N+7
```

```
      SAFE=SAFETYS/ABS(X(I))
      IF(SAFE.LE.SAFMAR)GO TO 20
      IF(IT.EQ.1)RETURN
      IF(N.EQ.9)RETURN
      N=9
      GO TO 1
20    WRITE(2,22)IELEM
22    FORMAT(1H0,I10)
      WRITE(2,21)(X(I),I=1,8)
21    FORMAT(1X,8E13.6)
      IF(IT.EQ.1)RETURN
      WRITE(2,21)(X(I),I=9,16)
      RETURN
      END
```

SUBROUTINE ELAS

Entry Point: ELAS

Purpose: To determine if the stress in an elastic spring element has reached the critical level.

Calling Sequence:

```
CALL ELAS
```

Blank Common: See Main Program

X(J) - not used

```
      SUBROUTINE ELAS
      COMMON ISAFT(10000),SAFMAR,SAFTVAL(1000,3)
      WRITE(2,3)
3     FORMAT(1H1," ELASTIC SPRINGS"/" ELEMENT    STRESS")
1     READ(3,2)A,IELEM
2     FORMAT(A10,I12)
      IF(A.EQ.10HNEW RECORD)GO TO 10
      READ(3,4)X
4     FORMAT(10X,F13.0)
      IF(X.GT.0.0)GO TO 5
      IF(ISAFT(IELEM).EQ.0.0)GO TO 1
      SAFETY=SAFTVAL(ISAFT(IELEM),2)
      SAFE=SAFETY/X
      IF(SAFE.GT.SAFMAR)GO TO 1
7     WRITE(2,6)IELEM,X
6     FORMAT(1X,I7,E13.6)
      GO TO 1
5     IF(ISAFT(IELEM).EQ.0.0)GO TO 1
      SAFETY=SAFTVAL(ISAFT(IELEM),1)
      SAFE=(SAFETY-X)/SAFETY
      IF(SAFE.LE.SAFMAR)GO TO 7
      GO TO 1
10    BACKSPACE 3
      RETURN
      END
```

SUBROUTINE CPLATE

Entry Point: CPLATE

Purpose: To determine if the stress in a plate element has reached the critical level.

Calling Sequence:

```
CALL CPLATE
```

External Reference: FAILURE

Blank Common: See Main Program

Z1	Z2	
X(1)	X(9)	Fibre Distance
X(2)	X(10)	Normal-X
X(3)	X(11)	Normal-Y
X(4)	X(12)	Shear-XY
X(5)	X(13)	O-Shear Angle
X(6)	X(14)	Major Principal
X(7)	X(15)	Minor Principal
X(8)	X(16)	Maximum Shear

```
      SUBROUTINE CPLATE
      COMMON ISAFT(10000),SAFMAR,SAFTVAL(1000,3),X(16)
      WRITE(2,3)
3     FORMAT(1H1," PLATE ELEMENTS"/" ELEMENT NUMBER"/
     12X,"FIBER DISTANCE",5X,"NORMAL-X",6X,"NORMAL-Y",6X,"SHEAR-XY",
     27X,"ANGLE",7X,"MAJPRIN",6X,"MINPRIN",2X,"MAX-SHEAR")
1     READ(3,2)A,IELEM
2     FORMAT(A10,I12)
      IF(A.EQ.10HNEW RECORD)GO TO 10
      DO 6 I=1,16
      READ(3,4)X(I)
4     FORMAT(10X,F13.0)
6     CONTINUE
      IT=2
      CALL FAILURE(IELEM,IT)
      GO TO 1
10    BACKSPACE 3
      RETURN
      END
```

SUBROUTINE CBEAM

Entry Point: CBEAM

Purpose: To determine if a BAR or BEAM element stresses have exceeded the critical level.

Calling Sequence:

```
CALL CBEAM(ITYPE)
```

ITYPE = 2 for a CBEAM element

34 for a CBAR element

Blank Common: See Main Program

X(1)	SA1	X(9)	SB1
X(2)	SA2	X(10)	SB2
X(3)	SA3	X(11)	SB3
X(4)	SA4	X(12)	SB4
X(5)	Axial	X(13)	SB-max
X(6)	SA-max	X(14)	SB-min
X(7)	SA-min	X(15)	Safety Margin in Compression
X(8)	Safety Margin in Tension		

```
      SUBROUTINE CBEAM(ITYPE)
      COMMON ISAFT(10000),SAFMAR,SAFTVAL(1000,3),X(15)
      DO 33 I=1,15
33    X(I)=0.0
      WRITE(2,3)
3     FORMAT(1H1," BEAM ELEMENTS"/" ELEMENT NUMBER"/
     16X,"SA1",10X,"SA2",10X,"SA3",10X,"SA4",9X,"AXIAL",
     28X,"SA-MAX",8X,"SA-MIN",2X,"SAFETY MARGIN IN TENSION"/
     36X,"SB1",10X,"SB2",10X,"SB3",10X,"SB4",8X,"SB-MAX",8X,
     4"SB-MIN",2X,"SAFETY MARGIN IN COMPRESSION")
1     READ(3,2)A,IELEM
2     FORMAT(A10,I12)
      IF(A.EQ.10HNEW RECORD)GO TO 10
      DO 6 I=1,15
      IF(I.EQ.12.AND.ITYPE.EQ.2)GO TO 6
      IF(I.EQ.4.AND.ITYPE.EQ.2)GO TO 6
      READ(3,4)X(I)
4     FORMAT(10X,F13.0)
6     CONTINUE
      IF(X(8).EQ.1.0)GO TO 15
      IF(X(15).EQ.1.0)GO TO 20
      IF(X(8).GT.SAFMAR.AND.X(15).GT.SAFMAR)GO TO 1
8     WRITE(2,7)IELEM,X
7     FORMAT(1H0,I10/1X,8E13.6/1X,7E13.6)
      GO TO 1
15    IF(X(15).GT.SAFMAR)GO TO 1
      IF(X(15).EQ.1.0)GO TO 1
      GO TO 8
20    IF(X(8).GT.SAFMAR)GO TO 1
      GO TO 8
10    BACKSPACE 3
      RETURN
      END
```

APPENDIX D

SAMPLE CONTROL CARD RECORDS

In order to use this package in conjunction with a NASTRAN analysis run with all the data on the input file the following control cards are required:

```
JOBCARD
ATTACH, NASTRAN, ···
COPYCR, INPUT, DATA1.
COPYCR, INPUT, DATA2.
NASTRAN (DATA1, DATA3, DUMMY) ATTACH
ATTACH, NASLIB, ···
LIBRARY, NASLIB.
STRESSA, DATA3.
YIELD, DATA1.
STRESSB, DATA2.
7/8/9
-NASTRAN DATA DECK -
7/8/9
-STRESSB DATA DECK -
6/7/8/9
```

Either DATA1 or DATA2 could be submitted from permanent file/tape by replacing the copy statements with appropriate ATTACH/REQUEST statements. If Failure Stress Data is to be input from existing files the YIELD, DATA1 card can be replaced with ATTACH/REQUEST or COPY statements to provide TAPE4 and TAPE5.

To generate a data file for later interactive use in STRESSB the last three statements of the control record should be replaced with.

```
REQUEST, TAPE3, *PF.
STRESSA.
CATALOG, TAPE3, tape3pfn, RP=999.
REQUEST, TAPE4, *PF.
REQUEST, TAPE5, *PF.
YIELD, DATA1.
CATALOG, TAPE4, tape4pfn, RP=999.
CATALOG, TAPE5, tape5pfn, RP=999.
```

If TAPE4 and TAPE5 are already available for this model the last 5 statements above can be omitted.

*U.S.Government Printing Office: 1977 — 757-080/220